
AN INTRODUCTION TO THE SAS MACRO LANGUAGE

Introduction

v a tool for extending and customizing the SAS System

v a tool for reducing the amount of text you must enter  to do common tasks

v can help you to produce better, more efficient,

v and more effective applications

v macro facility can be thought of as a programming system

v macro facility is a tool for manipulating SAS source statements

v The macro facility provides you with a flexible and powerful programming language for
manipulating the text of your SAS programs

v the macro facility allows you to work with two new objects

v macro variables (identified by the ampersand & )

v macros (identified by the percent sign % )

Macro Variables

v is a string of text identified by a name

v you create a macro variable by using a command

v you use it by typing its name where you want the string of text to appear

v the same macro variable can be used over and over again once it is defined



For example, lets say you have a master file called MASTER containing information on clients
that is updated periodically and that you keep a copy of the data prior to the last update in a file
called OLDMSTR. You may want to run a report program against either the new file or the old file.
You might approach this by creating a macro variable called dset and coding the following
statement:

%let dset = MASTER ;

Now, you can refer to the macro variable in your code, instead of using the actual data set name;
that is, you can code:

PROC PRINT DATA = &dset ;

The macro processor will substitute MASTER for &dset before it runs the source statement, so
your program will behave as if you coded:

PROC PRINT DATA = MASTER ;

v at this point, we don't have any significant savings

v its taken more statements to perform a simple PROC PRINT

v efficiency of using macros is

v you can use the macro variable &dset in many places

v you can change your source code in all of those places by changing the value of &dset in one
place

v this is a very common use of macro variables by themselves

v writing an extensive program and then customizing it for use on various data files by use of
macro variables



Example:

%let dset = MASTER ;

PROC PRINT DATA=&DSET  ;

RUN;

PROC FREQ DATA=&DSET ;

RUN;

DATA M_AND_F ;

   MERGE &DSET (IN = INMSTR)

  FINANCE (IN = INFIN);

  BY SUBSCR ;

  IF INMSTR AND INFIN ;

RUN ;

Referencing Macro Variables

v to substitute the value of a macro variable in your program, you must first reference it

v a macro variable reference

v is made by preceding the macro variable name with an ampersand (&)

v causes the macro processor to search for the named variable and  return the value if the variable
exists

v for example if the macro variable AMOUNT has a value of 800 :

        where cost > &amount ;

        generates

        where cost > 800 ;



v if you need to reference a macro variable within a literal  enclose the literal in double quotes.
v for example if the macro variable CITY has a value of Dallas :

        where c = "&city" ;

       generates

        where c = "Dallas" ;

        where c = '&city' ;

        generates

        where c = '&city' ;

v to monitor the value that is substituted when a macro variable is referenced, you can specify the
SYMBOLGEN system option

User-defined Macro Variables

v there are both user-defined and system-defined macro variables

v user defined macro variables are created with the %LET statement or are created within a
macro

v the %LET statement enables you to define a macro variable and  assign it a value

v the general form of the %LET statement is:

    %LET variable = value ;

v variable can be any name following SAS naming conventions

v value can be any string:

v the case of value is preserved

v maximum length is 32K characters

v minimum length is 0 characters(null value)



v numeric tokens are stored as character strings

v mathematical expressions are not evaluated

v quotes are stored as part of the value

v leading and trailing blanks are removed from value before the assignment is made

v if variable already exists, value replaces the current value

v Examples:

    %let name =  Joe Smith ;

    %let name = 'Joe Smith' ;

    %let start =;

    %let total = 0;

    %let sum = 3+4 ;

System-defined Macro Variables

v System-defined macro variables are created by the SAS Supervisor

v created at SAS invocation

v are global(always available)

v can be assigned values by the user in some cases

v Below are some of the system-defined macro variables

SYSDATE    date of SAS invocation

SYSDAY     day of the week of SAS invocation

SYSTIME    time of SAS invocation

SYSSCP     operation system being used

SYSVER     release of SAS system being used

SYSERR     return code set by last DATA or PROC step



SYSLAST    name of most recently created SAS data set in the form
           of libref.name. If no data set was created then the
           value is _NULL_ .

NOTE: the value of macro variables SYSDATE and SYSTIME are character strings, not SAS date
             or time values.

Example:

   %let job = UI00200 ;

   Footnote "&job run by Joe Programmer on &SYSDATE &SYSTIME";

   generates

   Footnote "UI00200 run by Joe Programmer on 06NOV95 15:45 "



Macros

v more complicated object than a macro variable

v like a macro variable, the end result of a macro is text that is included as a part of your SAS
program

v but unlike a macro variable, macros include executable statements that can manipulate
information and perform logical control

v macros and executable statements that are a part of  the macro processing language are all
identified by the percent character %

v the %let statement mentioned above is actually an executable macro statement that can appear
any place in a SAS program, not just in a macro

Defining and Calling Macros

v you define a macro by coding %macro as the first statement and %mend as the last statement

v you give the macro a name on the first statement and call it later by using that name

v for example, you could define a macro called %dset  that does the same thing as the &dset
variable mentioned

%MACRO DSET ;

   MASTER

%MEND ;

A statement using this macro to identify the file for a PROC PRINT could be:

PROC PRINT DATA = %DSET ;

v the result would be the same as the previous example

v you can add flexibility to this by defining macro parameters



Macro Parameters

v macro parameters are simply macro variables that are available within the macro and nowhere
else

v macro parameters provide a way to pass values to macro variables used within a macro

v for example, you could define the macro %dset as follows:

%MACRO DSET(FILE) ;

   &FILE

%MEND ;

v "FILE" is a macro parameter

v to run this macro you would code:

PROC PRINT DATA= %DSET(MASTER) ;

v the macro processor

v calls up the %dset macro

v defines the &file parameter in it

v substitutes the value MASTER



v  there are two type of macro parameters:   positional and keyword

v keyword parameters are those which start with the name of the macro variable followed by an
= symbol  and optionally followed by an initial value

v the general form is

%macro name_of_macro(positional.., keyword=Value);

 SAS Statements

 Macro statements

%mend ;

v positional parameters, if any, must appear first  in the parameter list

v positional parameters receive their values in the order (position) in which they are specified in
the macro invocation

v keyword parameters may appear in any order after positional parameters

v parameters are separated by commas

v Example:

%Macro means(procopt,vars,dsn=_last_) ;

   proc means &procopt data = &dsn;

   var &vars ;

%mend means;

v PROCOPT and VARS are positional parameters

v DSN is a keyword parameter

v the parameter DSN will be equal to _last_  unless  a different value is passed to it



v the command

  %means(mean t prt,educ sbp,dsn=one)

  will yield

 proc means mean t prt data=one;
    var educ sbp;

v and the command

 %means(n mean var,sbp)

 will yield

 proc means n mean var data=_last_;
   var sbp;

Macro Processing

v macro processing is triggered by the characters & and %

v when a & is found, the macro processor will search the macro variable for the value of the
variable in the string  immediately following

v the value is substituted for the reference to the macro variable

v when a % is encountered, the macro processor will interpret the following characters as a macro
that you define or as a macro function

v don't name your macros with the same name as a macro function

v when a reference to a macro is found, the macro will be run, and the result will be inserted in
place of  the reference



v after the macro processor has interpreted the value of a macro variable, a macro, or a macro
function, the  result will be scanned again until no more references with & or % exist

v the final result is treated as if it were a part of your original SAS program

Conditional Execution

v you can perform conditional execution at the macro level with

  %IF-%THEN and %ELSE statements

v the general form of %IF=%THEN and %ELSE statements is:

   %IF expression %THEN text ;

   %ELSE text ;

v expression can be any valid macro expression

v the %ELSE statement is optional

v text following the %THEN and %ELSE keywords can be:

v a macro programming statement

v constant text

v an expression

v a macro variable reference

v a macro call



v Use  %DO and %END statements following %THEN or %ELSE  to submit text that contains
semicolons

    %IF expression %THEN %DO ;

        statement ;

        statement ; ....

    %END ;

    %ELSE %DO ;

        statement ;

        statement ; ....

    %END ;

v NOTE: the macro statements listed above can only be used inside a macro definition



v Examples:

  %Macro Check(file) ;

    proc print data= &file ;

    var

    %if &file=master %then var1-var3 ;

      %else old1-old3 ;

    ;

    run;

  %Mend ;

  %Check(master) would resolve to:

   proc print data= master ;

    var var1-var3 ;

   run;

  check with any other value would resolve to:

   proc print data= master ;

    var old1-old3 ;

   run;



  %Macro reports(R1=YES,R2=YES) ;

    %if %upcase(R1)=YES %then %do;

      proc means data=one ;

         var x1-x10 ;

      title "report #1 run on &sysdate";

      run;

    %end;

    %if %upcase(R2)=YES %then %do;

      proc means data=two ;

         var y1-y10 ;

      title "report #2 run on &sysdate";

      run;

    %end;

  %Mend;

  %Reports(R1=NO,R2=YES) would resolve to:

  proc means data=two ;

    var y1-y10 ;

    title "report #2 run on 09NOV95";

  run;

v You can use the SAS system options MPRINT & MLOGIC to monitor  macro execution

v MPRINT prints the text sent to the SAS compiler as a result of  macro execution

v MLOGIC prints messages indicating macro actions taken during  macro execution



Macro Functions

v many macro applications require character manipulation

v below are a list of selected macro character functions

FUNCTION NAME                                      DESCRIPTION

%UPCASE translates letters from lower case to upper case
%LENGTH determines the length of a character string
%SUBSTR produces a substring of a character string
%SCAN extracts a word from a character string
%INDEX finds the position of the first occurrence of a character

string within another string

%UPCASE

v most macro comparisons are case sensitive

v the %UPCASE function translates all letters in an argument  to uppercase

  %Macro Check(value=) ;

    %if %upcase(&value)=FEMALE %then %do ;

      proc print data=one ;

      run;

    %end ;

  %Mend ;

  In the macro call  %Check(value=female) the macro variable VALUE resolves to FEMALE. If
  the %UPCASE function was omitted VALUE would resolve to female .



%SUBSTR

v general form of the %SUBSTR function is:

  %SUBSTR(argument,position,n)

v the first two arguments are required

v returns the portion of argument beginning at position for a length of n

v argument, position, and n values can contain:

v constant text

v macro variable references

v macro functions

v macro calls



%SCAN

v general form of the %SCAN function is:

  %SCAN(argument,n,delimiter)

v the first two arguments are required

v returns the nth word of argument, where words are strings of  characters separated by one or
more delimiters

v returns a null string if there are fewer than n words in argument

v argument and n values can contain:

v constant text

v macro variable references

v macro functions

v macro calls

v Example:

   %MACRO MODELS(value=) ;

      %let var1 = %scan(&value,1) ;

      %let var2 = %scan(&value,2) ;

      proc reg ;

         model y=&var1 &var2 ;

      run;

    %MEND ;



    %MODELS(value=educ sbp)  would resolve to:

      proc reg ;

         model y=educ sbp ;

      run;

Writing Text to the SAS Log

v when errors or problems occur in macro processing you may want to write your own messages
to the SAS log

v the %PUT statement writes messages to the SAS log

v the general form of the %PUT statement is:

   %PUT text ;

v the %PUT statement:

v can be used inside or outside a macro definition

v writes to the SAS log only

v always writes to a new log line starting in column one

v resolves macro triggers in text before text is written

v does not require quotes around text

v Example

   %MACRO TEST(val=) ;

      %put the value of val is &val ;

   %MEND ;



   the macro call %test(sbp) puts the following line in the SAS log:

   the value of val is sbp

Iterative Processing

v many macro applications require iterative processing

v with the iterative %DO statement you can

v execute macro programming code

v generate SAS code

v For example:

   %MACRO DOLOOP ;

     %do i= %to 3 ;

         proc freq ;

            tables var&i ;

         run;

     %end ;

   %MEND ;

   %DOLOOP generates :

   proc freq ; tables var1 ; run;

   proc freq ; tables var2 ; run;

   proc freq ; tables var3 ; run;



v the general form of the iterative %DO statement is:

  %DO index-variable = start %TO stop %BY increment ;

      text ;

  %END ;

v the %DO and %END are valid only inside a macro definition

v the index-variable is a macro variable

v the index-variable is created if it doesn't already exist

v the start, stop, and increment can be any valid expressions that resolve to integers



v Example:

%LET CAUSES = CHD CVD CANCER ;

%MACRO RATE ;

   %do i=1 %to 3;

       %let cause = %scan(&causes,&i) ;

       proc logist ;

          model &cause = age bmi sbp ;

       run;

   %end;

%MEND ;

the macro call %RATE resolves to:

       proc logist ;

          model chd = age bmi sbp ;

       run;

        proc logist ;

          model cvd = age bmi sbp ;

       run;

        proc logist ;

          model cancer = age bmi sbp ;

       run;



v Example:  run a report for a specified range of years

%MACRO REPORT(start=,end=) ;

   %do i=&start %to &end %by 1  ;

       proc freq ;

         tables x1 x2 x3 ;

         where year = &i ;

         title "report for year &i" ;

   %end ;

%MEND ;

The macro call %REPORT(start=1990,end=1992) would run the report for years 1990, 1991, and
1992.

Conditional Loops

v You can perform conditional iteration in macros with %DO %WHILE  and %DO %UNTIL
statements

v the general form of these statements are:

  %DO %WHILE(expression) ;

      text ;

  %END ;

  %DO %UNTIL(expression) ;

      text ;

  %END ;

v expression can be any valid macro expression



 %DO %WHILE loops

v evaluate expressions at the top of the loop

v execute repetitively while expression is true

 %DO %UNTIL loops

v evaluate expressions at the bottom of the loop

v executes repetitively until expression is true

v execute at least once

v Example:

%LET CAUSES = CHD CVD CANCER ;

%MACRO RATE ;

   %let i=1 ;

   %do %until (%scan(&causes,&i) = ) ;

       %let cause = %scan(&causes,&i) ;

       proc logist ;

          model &cause = age bmi sbp ;

       run;

       %let i=%eval(&i + 1) ;

   %end;

%MEND ;



the macro call %RATE resolves to:

       proc logist ;

          model chd = age bmi sbp ;

       run;

        proc logist ;

          model cvd = age bmi sbp ;

       run;

        proc logist ;

          model cancer = age bmi sbp ;

       run;



Data Step Interface

v in many applications you need to create macro variables that are assigned values based on data
values

v you can use the SYMPUT routine to assign a value available  to the DATA step to a macro
variable

v general form of the SYMPUT routine is:

  CALL SYMPUT(macro-variable,text) ;

v macro-variable is assigned the character value of text

v you can specify macro-variable or text as a

v literal

v DATA step variable

v DATA step expression


